
G02 – Correlation and Regression Analysis

G02DAF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

G02DAF performs a general multiple linear regression when the independent variables may be linearly
dependent. Parameter estimates, standard errors, residuals and influence statistics are computed.
G02DAF may be used to perform a weighted regression.

2 Specification

SUBROUTINE G02DAF(MEAN, WEIGHT, N, X, LDX, M, ISX, IP, Y, WT, RSS,
1 IDF, B, SE, COV, RES, H, Q, LDQ, SVD, IRANK, P,
2 TOL, WK, IFAIL)
INTEGER N, LDX, M, ISX(M), IP, IDF, LDQ, IRANK, IFAIL
real X(LDX,M), Y(N), WT(∗), RSS, B(IP), SE(IP),
1 COV(IP∗(IP+1)/2), RES(N), H(N), Q(LDQ,IP+1),
2 P(2∗IP+IP∗IP), TOL, WK(5∗(IP-1)+IP∗IP)
LOGICAL SVD
CHARACTER∗1 MEAN, WEIGHT

3 Description

The general linear regression model is defined by:

y = Xβ + ε

where y is a vector of n observations on the dependent variable,
X is a n by p matrix of the independent variables of column rank k,
β is a vector of length p of unknown parameters,

and ε is a vector of length n of unknown random errors such that var ε = V σ2, where V is a
known diagonal matrix.

If V = I, the identity matrix, then least-squares estimation is used. If V �= I, then for a given weight
matrix W ∝ V −1, weighted least-squares estimation is used.

The least-squares estimates β̂ of the parameters β minimize (y − Xβ)T (y − Xβ) while the weighted
least-squares estimates minimize (y − Xβ)T W (y − Xβ).

G02DAF finds a QR decomposition of X (or W 1/2X in weighted case), i.e.,

X = QR∗ (or W 1/2X = QR∗)

where R∗ =
(

R
0

)
and R is a p by p upper triangular matrix and Q is an n by n orthogonal matrix. If R

is of full rank, then β̂ is the solution to:
Rβ̂ = c1

where c = QT y (or QT W 1/2y) and c1 is the first p elements of c. If R is not of full rank a solution is
obtained by means of a singular value decomposition (SVD) of R,

R = Q∗

(
D 0
0 0

)
PT ,

where D is a k by k diagonal matrix with non-zero diagonal elements, k being the rank of R and Q∗ and
P are p by p orthogonal matrices. This gives the solution

β̂ = P1D
−1QT

∗1
c1

P1 being the first k columns of P , i.e., P = (P1P0) and Q∗1
being the first k columns of Q∗.

Details of the SVD, are made available, in the form of the matrix P ∗:

P ∗ =
(

D−1PT
1

PT
0

)
.

This will be only one of the possible solutions. Other estimates may be obtained by applying constraints
to the parameters. These solutions can be obtained by using G02DKF after using G02DAF. Only certain
linear combinations of the parameters will have unique estimates, these are known as estimable functions.

The fit of the model can be examined by considering the residuals, ri = yi − ŷ, where ŷ = Xβ̂ are the
fitted values. The fitted values can be written as Hy for an n by n matrix H . The ith diagonal elements
of H , hi, give a measure of the influence of the ith values of the independent variables on the fitted
regression model. The values hi are sometimes known as leverages. Both ri and hi are provided by
G02DAF.

The output of G02DAF also includes β̂, the residual sum of squares and associated degrees of freedom,
(n − k), the standard errors of the parameter estimates and the variance-covariance matrix of the
parameter estimates.

In many linear regression models the first term is taken as a mean term or an intercept, i.e., Xi,1 = 1, for
i = 1, 2, . . . , n. This is provided as an option. Also only some of the possible independent variables are
required to be included in a model, a facility to select variables to be included in the model is provided.

Details of the QR decomposition and, if used, the SVD, are made available. These allow the regression
to be updated by adding or deleting an observation using G02DCF, adding or deleting a variable using
G02DEF and G02DFF or estimating and testing an estimable function using G02DNF.

Edition), Baltimore

[4] Hammarling S (1985) The singular value decomposition in multivariate statistics SIGNUM Newsl.
20 (3) 2–25

[5] McCullagh P and Nelder J A (1983) Generalized Linear Models Chapman and Hall

[6] Searle S R (1971) Linear Models Wiley

5 Parameters

1: MEAN — CHARACTER*1 Input

On entry: indicates if a mean term is to be included.

If MEAN = ’M’ (Mean), a mean term, intercept, will be included in the model.
If MEAN = ’Z’ (Zero), the model will pass through the origin, zero-point.

Constraint: MEAN = ’M’ or ’Z’.

2: WEIGHT — CHARACTER*1 Input

On entry: indicates if weights are to be used.

If WEIGHT = ’U’ (Unweighted), least-squares estimation is used.
If WEIGHT = ’W’ (Weighted), weighted least-squares is used and weights must be supplied
in WT.

Constraint: WEIGHT = ’U’ or ’W’.

3: N — INTEGER Input

On entry: the number of observations, n.

Constraint: N ≥ 2.

4: X(LDX,M) — real array Input

On entry: X(i, j) must contain the ith observation for the jth independent variable, for i =
1, 2, . . . , n; j = 1, 2, . . . , m.

5: LDX — INTEGER Input

On entry: the first dimension of the array X as declared in the (sub)program from which G02DAF
is called.

Constraint: LDX ≥ N.

6: M — INTEGER Input

On entry: the total number of independent variables in the data set, m.

Constraint: M ≥ 1.

7: ISX(M) — INTEGER array Input

On entry: indicates which independent variables are to be included in the model.

If ISX(j) > 0, then the variable contained in the jth column of X is included in the regression model.

Constraints:

ISX(j) ≥ 0, for j = 1, 2, . . . , m.
If MEAN = ’M’, then exactly IP− 1 values of ISX must be > 0.
If MEAN = ’Z’, then exactly IP values of ISX must be > 0.

8: IP — INTEGER Input

On entry: the number of independent variables in the model, including the mean or intercept if
present.

Constraint: 1 ≤ IP ≤ N.

9: Y(N) — real array Input

On entry: observations on the dependent variable, y.

10: WT(∗) — real array Input

On entry: if WEIGHT = ’W’, then WT must contain the weights to be used in the weighted
regression.

If WT(i) = 0.0, then the ith observation is not included in the model, in which case the effective
number of observations is the number of observations with non-zero weights. The values of
RES and H will be set to zero for observations with zero weights.
If WEIGHT = ’U’, then WT is not referenced and the effective number of observations is n.

Constraint: if WEIGHT = ’W’, WT(i) ≥ 0.0, for i = 1, 2, . . . , n.

11: RSS — real Output

On exit: the residual sum of squares for the regression.

12: IDF — INTEGER Output

On exit: the degrees of freedom associated with the residual sum of squares.

13: B(IP) — real array Output

On exit: B(i), i = 1, 2, . . . ,IP contains the least-squares estimates of the parameters of the regression
model, β̂.

If MEAN = ’M’, then B(1) will contain the estimate of the mean parameter and B(i + 1)
will contain the coefficient of the variable contained in column j of X, where ISX(j) is the ith
positive value in the array ISX.
If MEAN = ’Z’, then B(i) will contain the coefficient of the variable contained in column j of
X, where ISX(j) is the ith positive value in the array ISX.

14: SE(IP) — real array Output

On exit: SE(i), i = 1, 2, . . . , IP contains the standard errors of the IP parameter estimates given in
B.

15: COV(IP∗(IP+1)/2) — real array Output

On exit: the first IP×(IP+1)/2 elements of COV contain the upper triangular part of the variance-
covariance matrix of the IP parameter estimates given in B. They are stored packed by column, i.e.,
the covariance between the parameter estimate given in B(i) and the parameter estimate given in
B(j), j ≥ i, is stored in COV(j × (j − 1)/2 + i).

16: RES(N) — real array Output

On exit: the (weighted) residuals, ri, for i = 1, 2, . . . , n.

17: H(N) — real array Output

On exit: the diagonal elements of H , hi, for i = 1, 2, . . . , n.

18: Q(LDQ,IP+1) — real array Output

On exit: the results of the QR decomposition:

the first column of Q contains c,
the upper triangular part of columns 2 to IP + 1 contain the R matrix,
the strictly lower triangular part of columns 2 to IP + 1 contain details of the Q matrix.

19: LDQ — INTEGER Input

On entry: the first dimension of the array Q as declared in the (sub)program from which G02DAF
is called.

Constraint: LDQ ≥ N.

20: SVD — LOGICAL Output

On exit: if a singular value decomposition has been performed then SVD will be .TRUE., otherwise
SVD will be .FALSE..

21: IRANK — INTEGER Output

On exit: the rank of the independent variables.

If SVD = .FALSE., then IRANK = IP.
If SVD = .TRUE., then IRANK is an estimate of the rank of the independent variables.

IRANK is calculated as the number of singular values greater that TOL × (largest singular value).
It is possible for the SVD to be carried out but IRANK to be returned as IP.

22: P(2∗IP+IP∗IP) — real array Output

On exit: details of the QR decomposition and SVD if used.

If SVD = .FALSE., only the first IP elements of P are used, these will contain the zeta values
for the QR decomposition (see F08AEF for details).
If SVD = .TRUE., the first IP elements of P will contain the zeta values for the QR
decomposition (see F08AEF for details) and the next IP elements of P contain singular values.
The following IP by IP elements contain the matrix P∗ stored by columns.

23: TOL — real Input

On entry: the value of TOL is used to decide if the independent variables are of full rank and if not
what is the rank of the independent variables. The smaller the value of TOL the stricter the criterion
for selecting the singular value decomposition. If TOL = 0.0, then the singular value decomposition
will never be used, this may cause run time errors or inaccurate results if the independent variables
are not of full rank.

Suggested value: TOL = 0.000001.

Constraint: TOL ≥ 0.0.

24: WK(5∗(IP−1)+IP∗IP) — real array Output

On exit: if on exit SVD = .TRUE., then WK contains information which is needed by G02DGF,
otherwise WK is used as workspace.

25: IFAIL — INTEGER Input/Output

On entry: IFAIL must be set to 0, −1 or 1. For users not familiar with this parameter (described
in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors detected by the routine:

IFAIL = 1

On entry, N < 2,

or M < 1,

or LDX < N,

or LDQ < N,

or TOL < 0.0,

or IP ≤ 0,
or IP > N.

IFAIL = 2

On entry, MEAN �= ’M’ or ’Z’,
or WEIGHT �= ’W’ or ’U’.

IFAIL = 3

On entry, WEIGHT = ’W’ or ’V’ and a value of WT < 0.0.

IFAIL = 4

On entry, a value of ISX < 0,

or the value of IP is incompatible with the values of MEAN and ISX,

or IP is greater than the effective number ofobservations.

IFAIL = 5

The degrees of freedom for the residuals are zero, i.e., the designated number of parameters is equal
to the effective number of observations. In this case the parameter estimates will be returned along
with the diagonal elements of H , but neither standard errors nor the variance-covariance matrix
will be calculated.

IFAIL = 6

The singular value decomposition has failed to converge, see F02WUF. This is an unlikely error.

7 Accuracy

The accuracy of this routine is closely related to the accuracy of F08AEF and F02WUF. These routine
documents should be consulted.

8 Further Comments

Standardised residuals and further measures of influence can be computed using G02FAF. This routine
requires, in particular, the results stored in RES and H.

9 Example

Data from an experiment with four treatments and three observations per treatment are read in. The
treatments are represented by dummy (0 − 1) variables. An unweighted model is fitted with a mean
included in the model.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* G02DAF Example Program Text
* Mark 14 Release. NAG Copyright 1989.
* .. Parameters ..

INTEGER MMAX, NMAX
PARAMETER (MMAX=5,NMAX=12)
INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)

* .. Local Scalars ..
real RSS, TOL
INTEGER I, IDF, IFAIL, IP, IRANK, J, M, N
LOGICAL SVD
CHARACTER MEAN, WEIGHT

* .. Local Arrays ..
real B(MMAX), COV((MMAX*MMAX+MMAX)/2), H(NMAX),

+ P(MMAX*(MMAX+2)), Q(NMAX,MMAX+1), RES(NMAX),
+ SE(MMAX), WK(MMAX*MMAX+5*(MMAX-1)), WT(NMAX),
+ X(NMAX,MMAX), Y(NMAX)
INTEGER ISX(MMAX)

* .. External Subroutines ..
EXTERNAL G02DAF

* .. Executable Statements ..
WRITE (NOUT,*) ’G02DAF Example Program Results’

* Skip heading in data file
READ (NIN,*)
READ (NIN,*) N, M, WEIGHT, MEAN
WRITE (NOUT,*)
IF (N.LE.NMAX .AND. M.LT.MMAX) THEN

IF (WEIGHT.EQ.’W’ .OR. WEIGHT.EQ.’w’) THEN
DO 20 I = 1, N

READ (NIN,*) (X(I,J),J=1,M), Y(I), WT(I)
20 CONTINUE

ELSE
DO 40 I = 1, N

READ (NIN,*) (X(I,J),J=1,M), Y(I)
40 CONTINUE

END IF
READ (NIN,*) (ISX(J),J=1,M)

* Calculate IP
IP = 0
IF (MEAN.EQ.’M’ .OR. MEAN.EQ.’m’) IP = IP + 1
DO 60 I = 1, M

IF (ISX(I).GT.0) IP = IP + 1
60 CONTINUE

* Set tolerance
TOL = 0.00001e0
IFAIL = 0

*
CALL G02DAF(MEAN,WEIGHT,N,X,NMAX,M,ISX,IP,Y,WT,RSS,IDF,B,SE,

+ COV,RES,H,Q,NMAX,SVD,IRANK,P,TOL,WK,IFAIL)
*

IF (SVD) THEN
WRITE (NOUT,99999) ’Model not of full rank, rank = ’, IRANK
WRITE (NOUT,*)

END IF
WRITE (NOUT,99998) ’Residual sum of squares = ’, RSS
WRITE (NOUT,99999) ’Degrees of freedom = ’, IDF
WRITE (NOUT,*)
WRITE (NOUT,*) ’Variable Parameter estimate Standard error’
WRITE (NOUT,*)
DO 80 J = 1, IP

WRITE (NOUT,99997) J, B(J), SE(J)
80 CONTINUE

WRITE (NOUT,*)
WRITE (NOUT,*) ’ Obs Residuals H’
WRITE (NOUT,*)
DO 100 I = 1, N

WRITE (NOUT,99997) I, RES(I), H(I)
100 CONTINUE

END IF
STOP

*
99999 FORMAT (1X,A,I4)
99998 FORMAT (1X,A,e12.4)
99997 FORMAT (1X,I6,2e20.4)

END

9.2 Program Data

G02DAF Example Program Data
12 4 ’U’ ’M’
1.0 0.0 0.0 0.0 33.63
0.0 0.0 0.0 1.0 39.62
0.0 1.0 0.0 0.0 38.18
0.0 0.0 1.0 0.0 41.46
0.0 0.0 0.0 1.0 38.02
0.0 1.0 0.0 0.0 35.83
0.0 0.0 0.0 1.0 35.99
1.0 0.0 0.0 0.0 36.58
0.0 0.0 1.0 0.0 42.92
1.0 0.0 0.0 0.0 37.80
0.0 0.0 1.0 0.0 40.43
0.0 1.0 0.0 0.0 37.89
1 1 1 1

9.3 Program Results

G02DAF Example Program Results

Model not of full rank, rank = 4

Residual sum of squares = 0.2223E+02
Degrees of freedom = 8

Variable Parameter estimate Standard error

1 0.3056E+02 0.3849E+00
2 0.5447E+01 0.8390E+00
3 0.6743E+01 0.8390E+00
4 0.1105E+02 0.8390E+00
5 0.7320E+01 0.8390E+00

Obs Residuals H

1 -0.2373E+01 0.3333E+00
2 0.1743E+01 0.3333E+00
3 0.8800E+00 0.3333E+00
4 -0.1433E+00 0.3333E+00
5 0.1433E+00 0.3333E+00
6 -0.1470E+01 0.3333E+00
7 -0.1887E+01 0.3333E+00
8 0.5767E+00 0.3333E+00
9 0.1317E+01 0.3333E+00

10 0.1797E+01 0.3333E+00
11 -0.1173E+01 0.3333E+00
12 0.5900E+00 0.3333E+00

